Background on the competition model

Consider two species, say sheep and cows, competing for limited food supplies. Denote the population of sheep by s and cows by c. In the absence of cows, the growth rate of sheep population is given by the logistic form $r_s(1 - s/L_s)$. When cows are present, the growth rate decreases by $k_s\, c$. Thus the sheep population s satisfies

$$s' = s\left(r_s - \frac{r_s}{L_s} s - k_s c\right). \tag{1}$$

Similarly, the cow population c satisfies

$$c' = c\left(r_c - \frac{r_c}{L_c} c - k_c s\right). \tag{2}$$

Eqs. (1--2) is a 2D autonomous system. Below we plot the phase portrait for a particular set of parameters.

Phase portrait for $\begin{cases} s' = s(2 - s - c) \\ c' = c(3 - c - 2\,s) \end{cases}$

Define the ODE right hand side function:

```
In[1]:= P[s_, c_] := s (2 - s - c);
Q[s_, c_] := c (3 - c - 2 s);
```

The vector $(P(s, c), Q(s, c))$ generally has different lengths depending on (s, c), which makes the vector field difficult to visualize. In drawing the phase portrait, though, only the direction of this vector matters, not the length. Thus, to ease visualization, we plot the rescaled vector field $(P, Q)\Big/\sqrt{P^2 + Q^2}$:

In[3]:=
```
VectorPlot[{P[x, y] / Sqrt[P[x, y]^2 + Q[x, y]^2], Q[x, y] / Sqrt[P[x, y]^2 + Q[x, y]^2]},
  {x, 0, 2}, {y, 0, 3}, VectorPoints → Fine]
```

Power::infy : Infinite expression $\frac{1}{0.}$ encountered. ≫

Infinity::indet : Indeterminate expression 0. ComplexInfinity encountered. ≫

Out[3]=



You should be able to draw by hand the vertical nullcline (defined by $P(s, c) = 0$) and the horizontal nullcline (defined by $Q(s, c) = 0$), and label the arrow directions on these nullclines. Subsequently, you can verify that your hand drawing agrees with the *Mathematica* output below, where these nullclines are respectively shown in red and green:

In[4]:=
```
Show[VectorPlot[{P[x, y] / Sqrt[P[x, y]^2 + Q[x, y]^2], Q[x, y] / Sqrt[P[x, y]^2 + Q[x, y]^2]},
   {x, 0, 2}, {y, 0, 3}, VectorPoints → Fine],
  ContourPlot[P[x, y] == 0, {x, -1, 4}, {y, -1, 4}, ContourStyle → Red,
   PlotPoints → 100], ContourPlot[Q[x, y] == 0, {x, -1, 4},
   {y, -1, 4}, ContourStyle → Green, PlotPoints → 100]]
```

Power::infy : Infinite expression $\frac{1}{0.}$ encountered. ≫

Infinity::indet : Indeterminate expression 0. ComplexInfinity encountered. ≫

Out[4]=



Solve the ODE numerically up to $t = 10$, with initial condition $(s, c)(t = 0) = (s_0, c_0)$:

In[5]:=
```
SolveIVP[s0_, c0_] := NDSolve[{s'[t] == P[s[t], c[t]],
    c'[t] == Q[s[t], c[t]], s[0] == s0, c[0] == c0}, {s, c}, {t, 0, 10}];
```

Plot the vector field together with the numerical solutions (shown in red) obtained using different initial conditions:

In[6]:=
```
Show[VectorPlot[{P[x, y] / Sqrt[P[x, y]^2 + Q[x, y]^2], Q[x, y] / Sqrt[P[x, y]^2 + Q[x, y]^2]},
   {x, 0, 2}, {y, 0, 3}, VectorPoints → Fine],
  Table[ParametricPlot[Evaluate[{s[t], c[t]} /. SolveIVP[sc⟦1⟧, sc⟦2⟧]],
    {t, 0, 10}, PlotStyle → Red], {sc, {{0.1, 0.2}, {0.3, 0.1}, {1.5, 3}, {2, 2}}}]]
```

Power::infy : Infinite expression $\frac{1}{0.}$ encountered. ≫

Infinity::indet : Indeterminate expression 0. ComplexInfinity encountered. ≫

Out[6]=



The intersections between the horizontal and vertical nullclines yield four equilibria: $(s, c) = (0, 0)$ (unstable), $(0, 3)$ (stable), $(2, 0)$ (stable), $(1, 1)$ (unstable). It can be seen that depending on the initial condition, the system evolves into either of the two stable equilibria. This implies that in the long run, either sheep or cows will go extinct.